

INFO-H-509 : Technologies XML

TP 4 - Correction

Professeur : Stijn Vansummeren

Assistant : Michaël Waumans

<http://cs.ulb.ac.be/public/teaching/infh509>

2015 - 2016

Exercise 1.1

Donnez les types de publications dans le fichier.

```
<publications>
{
  for $type in distinct-values(/dblp/*/name())
  order by $type
  return
    <type>{$type}</type>
}
</publications>
```

Exercise 1.2

Donnez le nombre de publications pour chaque type.

```
<publications>
{
  for $type in distinct-values(/dblp/*/name())
  let $number := count(/dblp/*[name()=$type])
  order by $type
  return
    element reference
    {
      attribute type {$type},
      <number>{$number}</number>
    }
}
</publications>
```

Exercise 1.3

Donnez la liste des noms des auteurs.

```
<authors> {
  for $author in distinct-values(/dblp//author)
  order by $author
  return
    <name>{$author}</name>
}
</authors>
```

Exercise 1.4

Donnez le nombre d'auteurs.

```
let $nb_authors := count(distinct-values(/dblp//author))
return
  <nb_authors>{$nb_authors}</nb_authors>
```

Exercise 1.5

Donnez les noms des auteurs qui sont aussi éditeurs.

```
<authors_editors>
{
  for $author in distinct-values(/dblp//author)
  where exists (/dblp//editor[.=$author])
  order by $author
  return
    <name>{$author}</name>
}
</authors_editors>
```

Exercise 1.6

Donnez le nombre de publications par auteur.

```
for $author in distinct-values(/dblp//author)
let $number := count(/dblp/*[author=$author])
order by $author
return
  element author
  {
    <name>{$author}</name>,
    <number>{$number}</number>
  }
```

Exercise 1.7

Donnez les auteurs par ordre descendant du nombre de publications.

```
let $result := (
  for $author in distinct-values(/dblp//author)
  let $number := count(/dblp/*[author=$author])
  return
    element author
    {
      <name>{$author}</name>,
      <number>{$number}</number>
    }
)
for $i in $result
order by $i/number(number) descending
return
  $i
```

Exercise 1.8

Donnez le(s) auteur(s) qui ont le plus grand nombre de publications.

```
<prolific_authors>
{
  let $result := (
    for $author in distinct-values(/dblp//author)
    let $number := count(/dblp/*[author=$author])
    return
    element author
    {
      <name>{$author}</name>,
      <number>{$number}</number>
    }
  )
  let $max := max($result//number)
  for $i in $result[number=$max]
  return $i
}
</prolific_authors>
```

Exercise 1.9

Donnez pour chaque auteur, le nombre de types de publication.

```
for $author in distinct-values(/dblp//author)
let $refs := /dblp/*[author=$author]
let $number := count(distinct-values($refs/name()))
order by number($number) descending
return
element author
{
  <name>{$author}</name>,
  <pubtypes>{$number}</pubtypes>
}
```

Exercise 1.10

Donnez pour chaque auteur le nombre total de publications et le nombre de publications par type.

```
let $pubtypes := distinct-values(/dblp/*/name())
for $author in distinct-values(/dblp//author)
let $total := count(/dblp/*[author=$author])
order by $author
return
element author
{
  <name>{$author}</name>,
  <total>{$total}</total>,
  for $type in $pubtypes
  let $number := count(/dblp/*[name()=$type and author=$author])
  where $number > 0
  order by $type
  return
  element pubtype
  {
```

```

        attribute type {$type},
        <number>{$number}</number>
    }
}

```

Exercise 1.11

Donnez la liste des "proceedings" qui ont au moins un éditeur qui est aussi auteur d'au moins un article dans les "proceedings".

```

<proc_editor_author>
{
  for $proc in /dblp/proceedings
  where some $editor in $proc/editor satisfies
  exists (/dblp/inproceedings[crossref=$proc/@key]/author[. = $editor])
  return
    $proc/title
}
</proc_editor_author>

```

Exercise 1.12

Définissez une fonction qui imprime les références du type "article" au format HTML.

```

declare function local:format-article(
  $article as element(article)?)
as element()?
{
  let $authors :=
    if (not(empty($article/author))) then
      concat(string-join($article/author, ", "), ". ")
    else
      if (count($article/editor)>1) then
        concat(string-join($article/editor, ", "), ", editors. ")
      else
        concat(string-join($article/editor, ", "), ", editor. ")
  return
    <p> {
      concat($authors,$article/title," ",$article/journal," ",$article/volume,
if (not(empty($article/number))) then concat("(", $article/number, "): ") else "",
      if (not(empty($article/pages))) then concat($article/pages, ", ") else " ",
      $article/year, ".")
    } </p>
};
let $result := (
  for $article in /dblp/article
  return
    local:format-article($article)
)
return (
  <html>
    <head>
      <meta http-equiv="Content-Type" content="text/html; charset=utf-8"/>
      <title>DBLP Articles</title>
    </head>
    <body>
      <h2>DBLP Articles</h2>

```

```
        {$result}  
    </body>  
</html>  
)
```